



ساخت یک دمو تکنیکی 3 بعدی در سبک Racing توسط GM

بررسی قابلیت های نرم افزار Game Maker در توسعه بازیهای 3 بعدی

بهترین اندازه دید : 768 * 1024
نویسنده : علی کسای

Ali_nwdo@yahoo.com
WWW.Persian-Designers.COM

(برگرفته شده از دمو تکنیکی Nick Larin)

کلیه حقوق این مقاله برای نویسنده و سایت طراحان ایرانی محفوظ می باشد

مقدمه :

ساخت بازی های 3 بعدی جذاب و خیره کننده توسط موتور موجود در نرم افزار Game Maker به هیچ وجه غیر ممکن نیست... توابع موجود در نسخه 6.0 این نرم افزار ، بخوبی نیازهای اولیه ساخت یک بازی 3 بعدی زیبا را فراهم می آورند... از توابعی Projection اولیه گرفته تا الگوریتم های culling و 3d Collision داخلی که بازی ساز علاقمند و تازه کار را از درگیر شدن در مفاهیم پیچیده این مباحث باز میدارند و در عین حال کارایی قابل قبولی را نیز ارائه می دهند... در دمو تکنیکی که تا دقایقی دیگر ساخت آنرا شروع خواهیم کرد ، سعی شده تا از اکثر توابع موجود در ساختار نرم افزار استفاده شود... بهتر است فرض بر این گذاشته شود که مبلغ 20 دلار صرف خرید این نرم افزار شده است و حال می خواهیم حداکثر بهره برداری ممکن از هزینه را داشته باشیم 😊

همانند گذشته پیشنهاد میکنم که این مقاله را هنگامی که درگیری ذهنی خاصی ندارید مطالعه کنید... پخش یک موسیقی ملایم و نوشیدن یک فنجان چای نیز در درک بهتر مقاله مفید خواهد بود.

مطابق معمول ابتدا Sprite های بازی را در نرم افزار وارد میکنیم و مختصات آنها را با توجه به اعداد داده شده در جدول زیر تنظیم میکنیم... توجه داشته باشید که نام Sprite ها هنگام ورود به نرم افزار ، همان نام عادی فایلها میباشد...

نام تصویر	X Origin	Y Origin	Transparent	توضیحات
-----------	----------	----------	-------------	---------

wall1	16	4	True	
wall2	4	16	True	
spr_camera	8	8	True	
spr_cam_col	0	0	False	
spr_flame_light	36	31	False	
floor	0	0	False	
spr_smoke	15	15	True	
spr_system	0	0	True	
ceiling	0	0	False	
tex_floor	63	63	False	
tex_wall	0	0	True	
tex_corner	63	63	False	
spr_racer1	63	32	True	
spr_racer1t	64	32	True	
spr_racer1t	16	15	True	
spr_spd_bg	0	0	True	
spr_spd	0	0	True	

بعد از اتمام مرحله بالا ، تنها جلوه صوتی بازی تحت نام snd_thruster را نیز در آن وارد می کنیم... این جلوه صوتی نقش صدای موتور Racer ما را خواهد داشت...

حال نوبت به طراحی عناصر موجود در محیط بازی میرسد... تمامی Object های موجود در این دمو تکنیکی به همراه کلیه مشخصات در جدول زیر آورده شده اند... آنها را مطابق با اطلاعات موجود در این جدول یکی یکی ایجاد میکنیم :

* توجه : به حروف کوچک و بزرگ در نام گذاری عناصر دقت کنید...

نام Object	Sprite	Visible	Solid	Depth	Parent	Mask
------------	--------	---------	-------	-------	--------	------

obj_globals	spr_system	False	False	- 9999	-	-
camera	spr_camera	True	True	1	-	spr_cam_col
obj_player	spr_racer1	True	False	- 1	-	spr_camera
ceiling	ceiling	True	False	0	-	-
wall1	wall1	True	True	0	obj_solid	-
wall3	wall3	True	True	0	obj_solid	-
obj_2ddrawer	spr_system	True	False	- 9	-	-
obj_wall	floor	True	True	0	obj_solid	-
obj_smoke	spr_smoke	True	False	- 1	-	-
obj_solid	wall1	True	True	0	-	-
obj_road_hor	tex_floor	True	False	0	-	-

-	-	0	False	True	tex_floor	obj_road_ver
-	-	0	False	True	tex_corner	obj_road_c_down_left
-	-	0	False	True	tex_corner	obj_road_c_down_right
-	-	0	False	True	tex_corner	obj_road_c_up_left
-	-	0	False	True	tex_corner	obj_road_c_up_right

حال به برنامه نویسی برای عناصر فوق می پردازیم... کلیه حالات مطرح شده ، همان Event های همیشگی هستند که در مقالات قبلی بارها استفاده شدند... چنانچه جایی در ایجاد این Event ها مشکلی داشتید و یا در درک کاربرد آنها با مشکل مواجه شدید ، نگاهی به مقالات قبلی بیاندازید...

کدهای مربوط به obj_globals

: رویداد Create

```
//Player
global.player_spd = 0;
//Cars
global.car = 1;
global.car_maxspd[1] = 8;
global.car_accel[1] = 0.2;
global.car_turn[1] = 4;
```

کدهای مربوط به camera

: رویداد Create

```
d3d_start();
//sets the distant that the objects are visable at
d3d_set_fog(true,c_black,50,1000);
zoom=40;
x = obj_player.x-1;
y = obj_player.y;
```

: رویداد Step

```
cdir = point_direction(obj_player.x,obj_player.y,x,y);
speed = zoom + obj_player.speed*1;
direction = obj_player.dir;
x1=obj_player.x+hspeed;
y1=obj_player.y+vspeed;
direction = cdir;
x = obj_player.x+hspeed;
y = obj_player.y+vspeed;
x+= (x-x1)/8;
y+= (y-y1)/8;
cdir = direction;
//collision
//move_outside_solid(cdir,30);
```

رویداد Draw :

```
projection();
```

کدهای مربوط به obj_player

رویداد Create :

```
dir = 0;
friction = 0.1;
spd=0;
zpos = 0;
zspeed = 0;
zup = true;
```

رویداد Step :

```
instance_create(self.x+4-random(8),self.y+4-random(8),obj_smoke);
if speed > global.car_maxspd[global.car]
speed = global.car_maxspd[global.car];
if zup = true zspeed+=0.01; else zspeed+=-0.01;
zpos+=zspeed;
if zspeed > 0.1 zspeed = 0.1;
if zspeed < -0.1 zspeed = -0.1;
if zpos > 0.5 zup = false;
if zpos < -0.5 zup = true;
if keyboard_check(vk_up)
{
if not sound_isplaying(snd_thruster)
sound_loop(snd_thruster);
}
if not keyboard_check(vk_up) sound_stop(snd_thruster);
```

رویداد برخورد با obj_solid :

```
move_bounce_solid(false);
speed = speed/1.5;
```

رویداد Keyboard Key Left :

```
dir+=global.car_turn[global.car];
if dir > (360-global.car_turn[global.car]) dir = 0;
```

رویداد Keyboard Key Right :

```
dir+=-global.car_turn[global.car];
if dir < global.car_turn[global.car] dir = 360;
```

رویداد Keyboard Key Up :

```
motion_set(direction+dir,speed+global.car_accel[global.car]);
```

: رویداد Draw

```
dir2 = point_direction(y,x,camera.y,camera.x);
image_single = (dir2+dir)/(360/(image_number));
if image_single > image_number
image_single = image_single - image_number;
//FLAME LIGHT DRAW
if keyboard_check(vk_up)
{
d3d_transform_set_identity();
d3d_transform_set_scaling(1+random(0.2)+zpos/4,0.4,0.4);
d3d_transform_add_rotation_z(dir);
d3d_transform_add_translation(x,y,1);
draw_set_blend_mode(bm_add);
draw_sprite(spr_flame_light,-1,0,0);
draw_set_blend_mode(bm_normal);
d3d_transform_set_identity();
}
//SHADOW DRAW
{
d3d_transform_set_identity();
d3d_transform_set_scaling(0.6,0.6,0.6);
d3d_transform_add_rotation_z(dir);
d3d_transform_add_translation(x,y,1);
draw_sprite_ext(spr_shadow1,-1,0,0,1,1,0,0,0.4);
d3d_transform_set_identity();
}
//CAR DRAW
{
d3d_transform_set_identity();
d3d_transform_set_scaling(0.15,0.15,0.15);
d3d_transform_add_rotation_y(dir2);
d3d_transform_add_rotation_x(90);
d3d_transform_add_translation(x,y,6+zpos);
if keyboard_check(vk_up) draw_sprite(spr_racer1t,-1,0,0);
else draw_sprite(spr_racer1,-1,0,0);
d3d_transform_set_identity();
}
```

کدهای مربوط به ceiling

: رویداد Draw

```
draw_set_color(c_white);
//The size and texture used for the floor
d3d_draw_floor(0,0,32,800,600,32,sprite_get_texture(ceiling,0),5,5);
```

کدهای مربوط به wall1

رویداد Draw :

```
draw_set_color(c_white);  
d3d_draw_wall(x-16,y,0,x+16,y,15,sprite_get_texture(tex_wall,0),1,1);
```

کدهای مربوط به wall3

رویداد Draw :

```
draw_set_color(c_white);  
d3d_draw_wall(x,y-16,0,x,y+16,15,sprite_get_texture(tex_wall,0),1,1);
```

کدهای مربوط به obj_2ddrawer

رویداد Draw :

```
//global.car_maxspd[global.car] obj_player.speed 122  
spd=(122/global.car_maxspd[global.car])*obj_player.speed + 1;  
  
draw_set_color(c_white);  
d3d_set_projection_ortho(0,0,800,600,0);  
d3d_set_hidden(false);  
  
draw_text(20,20,'A Simple Racing Demo ( Persian-Designers )  
FPS: ' + string(fps));  
draw_sprite(spr_spd_bg,0,30,540);  
draw_sprite_part(spr_spd,0,0,0,spd,27,33,543);  
d3d_set_hidden(true);
```

کدهای مربوط به obj_wall

رویداد Draw :

```
draw_set_color(c_white);  
d3d_draw_wall(x-16,y,0,x+16,y,6,sprite_get_texture(tex_wall,0),1,1);
```

کدهای مربوط به obj_smoke

رویداد Create :

```
image_speed = 0.25;
```

```

rot = 5-random(10);
rot2 = 0;
size = 1;
trans = 0.2;
direction = random(360);
speed = 2;
friction = 0.1;
depth = -1-random(2);

```

: Animation End رویداد

```

with(self) instance_destroy();

```

: Draw رویداد

```

dir = point_direction(y,x,camera.y,camera.x);
rot2+=rot;
size+=0.02;
trans+=-0.0075;
{
    d3d_transform_set_identity();
    d3d_transform_set_scaling(0.15,0.15,0.15);
    d3d_transform_add_rotation_y(dir);
    d3d_transform_add_rotation_x(90);
    d3d_transform_add_translation(x,y,2);
    draw_set_blend_mode(bm_add);
    draw_sprite_ext(spr_smoke,-1,0,0,size,size,rot2,c_white,trans);
    draw_set_blend_mode(bm_normal);
    d3d_transform_set_identity();
}

```

obj_road_ver به کدهای مربوط به

: Create رویداد

```

image_angle = 90;
y+=1;

```

obj_road_c_down_right به کدهای مربوط به

: Create رویداد

```

image_angle = 90;
y+=1;

```

obj_road_c_up_left به کدهای مربوط به

رویداد Create :

```
image_angle = 270;
```

کدهای مربوط به obj_road_c_up_right

رویداد Create :

```
image_angle = 180;  
y+=1;  
x+=1;
```

کار برنامه نویسی اشیای موجود در بازی در اینجا به پایان میرسد... با این وجود هنوز یک مرحله دیگر از برنامه نویسی برنامه باقی مانده است... کدام مرحله ؟ درست حدس زدید... ایجاد اسکریپت **projection** که آنرا در رویداد Draw مربوط به Camera فراخوانی کردیم... بنابراین به ایجاد این تابع میپردازیم تا آخرین مرحله برنامه نویسی این دمو تکنیکی نیز به پایان برسد...

برای این منظور کافیهست تا یک اسکریپت جدید ایجاد کنید و کدهای زیر را در آن استفاده نمایید :

کدهای مربوط به اسکریپت projection

```
with(camera) d3d_set_projection(x,y,16,obj_player.x,obj_player.y,10,0,0,1);
```

بسیار خوب... کار برنامه نویسی این دمو تکنیکی در اینجا به پایان رسید...

طراحی یک ROOM برای بازی :


طبیعی است که طراحی یک اتاق (در واقع یک Path برای حرکت) کاری کاملاً سلیقه ای و دلخواه است... بنابراین هریک از شما میتوانید ROOM منحصر بفردی را ایجاد کنید... با این وجود طراحی اتاق در Game Maker و مخصوصاً در این دمو تکنیکی نیازمند در نظر داشتن چند شرط و محدودیت است که در زیر عنوان میشوند :

- 1) شما می بایست obj_globals , camera , obj_player را حتما در اتاق قرار بدهید.
- 2) برای دیوار های افقی (در نمای دو بعدی طراحی اتاق) باید از wall1 استفاده شود...
- 3) برای دیوار های عمودی (در نمای دو بعدی) باید از wall3 استفاده شود.
- 4) اگر از یک سیستم قدرتمند استفاده می کنید ، مقدار پیش فرض سرعت اتاق (30) را به مقداری بالاتر نظیر 50 افزایش دهید تا حرکت روانتری داشته باشید.
- 5) مکان قرار گیری obj_player ، همان مکان شروع بازی است...
- 6) ابعاد اتاق در اولین مرحله ساخت ، بهتر است بصورت 600 * 800 باشد. در غیر اینصورت در تنظیمات View ابتدا یک نمای جدید با ابعاد 800 در 600 تعریف کنید و سپس رزولوشن بازی را مساوی همین مقدار قرار دهید تا Stretch ناخواسته روی ندهد.

7) برای ترسیم صحیح خطوط جاده در پیچ ها دقت کنید. برای این منظور شما چهار جهت برای چهار نوع پیچ ممکن را در اختیار دارید که از قرار زیرند :


obj_road_c_down_left
obj_road_c_down_right
obj_road_c_up_left
obj_road_c_up_right

بسته به جهت چرخش پیچ میبایست از یکی از 4 فرم بالا استفاده کنید... تصویر زیر به درک بهتر موضوع کمک میکند :



فرم کلی اتاق در نهایت چیزی شبیه به شکل بالا می باشد... شما می توانید آنرا به دلخواه خود بزرگ یا کوچک کرده و سرعت اجرای آنرا تحت کنترل در بیاورید. میزان Fov را برای دستیابی به بهترین نتیجه ممکن مجدداً تنظیم کنید و از تکسچرهای جدیدی برای تزئین کف زمین و دیواره های جاده استفاده نمایید... در عین حال می توانید با تغییر پارامترهای دوربین موجود در بازی ، صحنه دید بازی را تغییر بدهید و نمای نشستن پشت فرمان را ایجاد کنید... یا اینکه به تقلید از بازی GTA ، یک رادپوی جمع و جور و کوچک برای ماشینتاز طراحی کنید تا چند آهنگ زیبا را برایتان پخش کند 😊

بازی شما بعد از اجرا چیزی شبیه تصویر زیر خواهد بود :



چنانچه توانسته باشید ساخت نمونه آموزشی این مقاله را با موفقیت به پایان برسانید ، مسلما از دیدن تصویر بالا لذت خواهد برد... با این وجود صرف کمی وقت و افزودن ظرافت و جزئیات بیشتر به کار ، جلوه و نمای کار شما را تا حد بسیار زیادی بالا می برد...

تصویری که در صفحه بعد ملاحظه می کنید ، نتیجه تنها 30 دقیقه کار اضافه بر ساختار بازی می باشد... مقایسه جزئیات و جلوه نهایی بازی را بر عهده خودتان می گذارم 😊



منتظر شنیدن نظرات و دیدن کارهای شما هستیم...

چنانچه از ساخت این بازی فارغ شده اید ، تامل نکنید و آنرا برای ما ارسال کنید... ما و سایر دوستان در تالارهای سایت از شما و بازیهایتان به گرمی استقبال خواهیم کرد و با هم در جهت بهبود و تکمیل آنها خواهیم کوشید... کافیت یکبار امتحان کنید...

<http://www.persian-designers.com/forum>

تالارهای گفتگو و تبادل نظر :


سایت طراحان ایرانی با هدف آموزش ساخت بازیهای کامپیوتری به زبان فارسی طراحی شده است و تا کنون مقالات متعددی در زمینه های مختلف برنامه نویسی و ساخت بازی در آن قرار گرفته است. مدیریت سایت از تمامی عزیزان علاقمند به بازی های کامپیوتری ، برنامه نویسان ، طراحان و سایر کسانی که به نحوی با بازی ها در ارتباطند ، دعوت به همکاری به عمل می آورد تا بدینوسیله یک پایگاه علمی و موثق در زمینه صنعت ساخت بازیهای کامپیوتری در ایران ایجاد گردد.

در ضمن بسیاری از نرم افزار های ساخت بازی های کامپیوتری که امروزه در سطح وسیع مورد استفاده قرار میگیرند ، در سایت جمع آوری شده است و با مبلغ بسیار ناچیزی در اختیار علاقمندان به طراحی بازی های کامپیوتری قرار داده شده است. استفاده از این نرم افزار ها در آغاز کار و به منظور آشنا شدن با اصول اولیه در طراحی بازیها بسیار موثر و مفید بوده و شما میتوانید تا با چند جستجوی ساده در این زمینه ، به صحت موضوع

پی ببرید، در زیر برخی از نرم افزار هایی که توسط فروشگاه الکترونیکی سایت به مشتاقان عرضه میشوند را ملاحظه می کنید :



مجموعه ذخایر ساخت بازی | مجموعه آموزشی ساخت بازی با Game Maker | نرم افزار 3D Game Studio



پکیج استاندارد ساخت بازیهای 3 بعدی | مجموعه موتورهای سورس باز 3 بعدی

آدرس فروشگاه الکترونیکی : <http://www.persian-designers.com/index.php?pid=1>

کلیه حقوق این مقاله برای نویسنده و سایت طراحان ایرانی (www.persian-designers.com) محفوظ می باشد
استفاده از مطالب موجود در این مقاله در صورت ذکر کامل منبع مجاز است